Solving partial differential equations on balls in R^n

Philip Korman and Dieter Schmidt
University of Cincinnati

This program solves the Dirichlet problem for semilinear Laplace equation on a unit ball in arbitrary space dimension, depending on a positive parameter λ:

(1) $\Delta u + \lambda f(u) = 0$ for $|x|<1$, $u = 0$ for $|x| = 1$ in n dimensions

Here Δ denotes the Laplacian. We compute positive solutions.

By the classical theorem of B. Gidas, W.-M. Ni and L. Nirenberg (Comm. Partial Differential Equations {\bf 6}, no. 8, pp. 883-901, 1979) positive solutions are are radially symmetric, so that u = u(r), where r= |x|, and hence u(r) satisfies an ODE

(2) $u_{rr} + \frac{n-1}{r} u_r + \lambda f(u) = 0, \quad u_r(0) = u(1) = 0.$

The theorem also asserts that $u_r < 0$ so that u(0) gives the maximal value of the solution u(r). Moreover, the value of u(0) is a global parameter, uniquely identifying the solution pair (λ ,u(r)), see the book of P. Korman (World Scientific, 2012). (If u(0)=5, there is a unique $\lambda=\lambda_0$ and the corresponding solution u_0(r), so that $u_0(0) = 5$.) It follows that two-dimensional curves in (λ ,u(0)) plane provide a complete picture of the solution set of the PDE (1) (or (2)).

We now describe the "shoot-and-scale" method for drawing the solution curves of (2). Perform "shooting", or solve the following initial-value problem

$$v_{rr} + \frac{n-1}{r} v_r + f(v) = 0, \quad v(0) = \alpha > 0, \; v_r(0) = 0.$$

Let ξ be the first root of v(r). Then u(r)=v(ξ r) provides the solution of (2) at $\lambda = \xi^2$, with u(0) = v(0) = α. This gives a point (ξ , α) on the solution curve. The program below computes 400 such points, then joins them to get a bifurcation curve. (Please execute the program.)

The program solves a Dirichlet problem on the ball and draws the bifurcation diagram

In[ ]:=
```
Clear["`*"]
f[u_] = u + 0.5 * u * Sin[u];
n = 3;
u00 = 0;
delu = 0.07;
nsteps = 400;
tend = 1000;
h = $MachineEpsilon;
dirichlet = {};
For[k = 1, k ≤ nsteps, k++,
  u0 = u00 + k * delu;
  sol = NDSolve[{x''[t] == -(n - 1) * x'[t] / t - f[x[t]], x[h] == u0, x'[h] == 0,
     WhenEvent[{x[t] * x'[t] == 0}, If[Abs[x[t]] < 10^(-8), AppendTo[dirichlet, {t^2, u0}]];
      "StopIntegration"]}, x, {t, h, tend}];
]

ListPlot[dirichlet, Joined → True, AxesLabel → {"λ", "u(0)"}, PlotRange → All, PlotStyle → Thickness[0.007],
 PlotLabel → "Bifurcation Diagram for Dirichlet problem"]
```

Out[ ]=



The points on the solution curve were recorded in the file named dirichlet. For example, the point (λ_0, α) with λ_0 = dirichlet[[100,1]] and α = dirichlet[[100,2]] lies on the solution curve. The corresponding solution u(r) is computed as follows.

In[ ]:=
```
λ0 = dirichlet⟦100, 1⟧;
α = dirichlet⟦100, 2⟧;
sol = NDSolveValue[{u''[r] + (n - 1) * u'[r] / r + λ0 f[u[r]] == 0, u[h] == α, u'[h] == 0}, u, {r, h, 1}];
Plot[sol[r], {r, h, 1}, PlotStyle → Thick]
```

Out[ ]=



Similar approach works for the p - Laplacian case, and for the Neumann problem . The Mathematica programs can be obtained from https://homepages.uc.edu/~kormanp/